U. S. FOREST SERVICE RESEARCH NOTE NE-80
1968

NE

FOREST SERVICE, U. S. DEPT. OF AGRICULTURE, 6816 MARKET STREET, UPPER DARBY, PA.

A COMPUTER PROGRAM FOR MAKING S-CONTRASTS INVOLVING LINEAR COMBINATIONS OF GROUP MEANS

Abstract. A description of a FORTRAN IV computer program that is used for making Scheffé's S-contrasts in one-way analyses.

Forestry research often involves experiments in which many treatments are tested. If the overall test of treatment equality results in a significant difference, it is usually necessary to make further tests involving linear combinations of the treatment responses. When many a *posteriori* comparisons are necessary and treatments have unequal sample sizes, the researcher often resorts to Scheffé's S-contrasts and finds himself involved in lengthy calculations.

Scheffé[1] describes the method of making contrasts for a one-way analysis. In short, the method results in the contrast $\sum_{1}^{I} c_i\beta_i$ with a confidence interval:

$$\sum_{1}^{I} c_i\beta_i \pm (I\text{-}1)(F_{\alpha;\ I-1,N-I})\ (s)\sqrt{\sum_{1}^{I} (c_i^2/J_i)}$$

Where:

c_i = a contrast multiplier for group i, and $\sum_{1}^{I} c_i = 0$.

β_i = arithmetic mean of group i.

[1] Scheffé, Henry. THE ANALYSIS OF VARIANCE. John Wiley, N. Y. 477 pp. 1959.

I $=$ number of groups.
F $=$ Fisher's F-value for I-1 and N-I degrees of freedom at the chosen (α) level of significance.
N $=$ total sample size (all groups combined).
s^2 $=$ the error mean square.
J_i $=$ sample size for group i.

By the same method, we may calculate an F-value for an individual contrast as:

$$F_{I-1,N-I} = \frac{\left(\sum_{1}^{I} c_i \beta_i \right)^2}{(I-1)s^2 \left(\sum_{1}^{I} c_i^2 \Big/ J_i \right)}$$

and we may check the F-value against a tabulated F at level of significance α.

The FORTRAN IV computer program described here is simply an automated method of making one-way analysis S-contrasts for linear combinations of group means with constant or varying sample sizes.

Description of Control Deck

The user must supply the following control deck as data to be operated on by the program:

Card No.	*Card Columns*	*Content*
1	1–4	Number of treatments or groups, right adjusted.
	5–8	Degrees of freedom for error, right adjusted.
	9–17	Error mean square, punched with decimal point.
	18–22	*MEANS* punched if individual treatment means are to be compared; blank otherwise.
2-I	1–10	Label for group.
(1 for each group)	11–13	Sample size, right adjusted
	14–20	Group total, punched with decimal point.
	21	Minus sign (—) if multiplier (c) for contrast is negative.

22–25	Multiplier (c) for contrast, punched with decimal point.
26–70	Repetition of format for columns 21–25 for each additional contrast to be made.

The program is limited to:

(1) A maximum of 10 contrasts (in addition to the contrasts of individual means).

(2) A maximum of 1,000 groups.

Example

If we assume that we have tested the effects of eight fertilizers on height growth of red pine seedlings and that the treatments are represented by unequal sample sizes, then the analysis might be:

Source	*df.*	*SS*	*M.S.*	*F.*
Fertilizer	7	94.0678	13.438	14.0
Error	134	128.7060	0.960	
Total	141	222.7738		

The treatment totals and their sample sizes are:

Treatment	J_i	*Total (growth in height)*
1	18	36.7
2	19	12.1
3	17	11.8
4	15	42.1
5	19	38.9
6	20	60.3
7	20	40.2
8	14	30.1

```
   8 134      0.960MEANS
TREAT. 1     18    36.7         .25    .25
TREAT. 2     19    12.1 .5    -.5
TREAT. 3     17    11.8 .5    -.5
TREAT. 4     15    42.1-.5            -.5
TREAT. 5     19    38.9         .25    .25
TREAT. 6     20    60.3-.5            -.5
TREAT. 7     20    40.2         .25    .25
TREAT. 8     14    30.1         .25    .25
```

Figure 1.—Control deck.

```
S CONTRASTS INVOLVING LINEAR COMBINATIONS OF ARITHMETIC MEANS -

EACH CONTRAST HAVING    7 AND  134 DEGREES OF FREEDOM

TREAT. 1    VS. TREAT. 2  , F= 0.2704E 01

TREAT. 1    VS. TREAT. 3  , F= 0.2353E 01

TREAT. 1    VS. TREAT. 4  , F= 0.7177E 00

TREAT. 1    VS. TREAT. 5  , F= 0.9890E-04

TREAT. 1    VS. TREAT. 6  , F= 0.1343E 01

TREAT. 1    VS. TREAT. 7  , F= 0.1177E-02

TREAT. 1    VS. TREAT. 8  , F= 0.1447E-01

TREAT. 2    VS. TREAT. 3  , F= 0.4380E-02

TREAT. 2    VS. TREAT. 4  , F= 0.5873E 01

TREAT. 2    VS. TREAT. 5  , F= 0.2813E 01

TREAT. 2    VS. TREAT. 6  , F= 0.8200E 01

TREAT. 2    VS. TREAT. 7  , F= 0.2734E 01

TREAT. 2    VS. TREAT. 8  , F= 0.2746E 01

TREAT. 3    VS. TREAT. 4  , F= 0.5292E 01

TREAT. 3    VS. TREAT. 5  , F= 0.2445E 01

TREAT. 3    VS. TREAT. 6  , F= 0.7366E 01

TREAT. 3    VS. TREAT. 7  , F= 0.2368E 01

TREAT. 3    VS. TREAT. 8  , F= 0.2422E 01

TREAT. 4    VS. TREAT. 5  , F= 0.7192E 00

TREAT. 4    VS. TREAT. 6  , F= 0.5536E-01

TREAT. 4    VS. TREAT. 7  , F= 0.8095E 00

TREAT. 4    VS. TREAT. 8  , F= 0.4647E 00

TREAT. 5    VS. TREAT. 6  , F= 0.1358E 01

TREAT. 5    VS. TREAT. 7  , F= 0.2025E-02

TREAT. 5    VS. TREAT. 8  , F= 0.1263E-01

TREAT. 6    VS. TREAT. 7  , F= 0.1503E 01

TREAT. 6    VS. TREAT. 8  , F= 0.9169E 00

TREAT. 7    VS. TREAT. 8  , F= 0.2402E-01

 0.5000E 00 X TREAT. 2
 0.5000E 00 X TREAT. 3
-0.5000E 00 X TREAT. 4
-0.5000E 00 X TREAT. 6
                    F= 0.1316E 02

 0.2500E 00 X TREAT. 1
-0.5000E 00 X TREAT. 2
-0.5000E 00 X TREAT. 3
 0.2500E 00 X TREAT. 5
 0.2500E 00 X TREAT. 7
 0.2500E 00 X TREAT. 8
                    F= 0.6871E 01

 0.2500E 00 X TREAT. 1
-0.5000E 00 X TREAT. 4
 0.2500E 00 X TREAT. 5
-0.5000E 00 X TREAT. 6
 0.2500E 00 X TREAT. 7
 0.2500E 00 X TREAT. 8
                    F= 0.2466E 01
```

Figure 2.—Program output.

```
$IBFTC STEST
      DIMENSION LABEL(1000,3),XJ(1000),TOT(1000),C(1000,10)
      READ(5,1)IGRPS,IDF,SQUARE,MEAN,S
    1 FORMAT(2I4,F9.0,A4,A1)
      DATA MEA,SN/4HMEAN,1HS/
      XGRPS=IGRPS
      DF=IDF
      IGRPDF=IGRPS-1
      DO 2 I=1,IGRPS
    2 READ(5,3)(LABEL(I,J),J=1,3),XJ(I),TOT(I),(C(I,J),J=1,10)
    3 FORMAT(2A4,A2,F3.0,F7.0,10F5.0)
      WRITE(6,4)IGRPDF,IDF
    4 FORMAT(85H1S CONTRASTS INVOLVING LINEAR COMBINATIONS OF ARITHMETIC
     1 MEANS - EACH CONTRAST HAVINGI5,4H ANDI5,20H DEGREES OF FREEDOM   )
      IF(MEA.NE.MEAN.OR.SN.NE.S)GO TO 7
      DO 5 I=1,IGRPDF
      IZ=I+1
      DO 5 K=IZ,IGRPS
      DIFF=TOT(I)/XJ(I)-TOT(K)/XJ(K)
      SUM=1./XJ(I)+1./XJ(K)
      F=(DIFF**2)/((XGRPS-1.)*SQUARE*SUM)
    5 WRITE(6,6)(LABEL(I,J),J=1,3),(LABEL(K,J),J=1,3),F
    6 FORMAT(2H0 2A4,A2,5H VS. 2A4,A2,4H, F=E11.4)
    7 DO 13 J=1,10
      WRITE(6,8)
    8 FORMAT(1H )
      DO 9 I=1,IGRPS
      IF(C(I,J).NE.0.0)GO TO 10
    9 CONTINUE
      GO TO 15
   10 DIFF=0.
      SUM=0.
      DO 12 I=1,IGRPS
      DIFF=DIFF+C(I,J)*(TOT(I)/XJ(I))
      IF(C(I,J).NE.0.)WRITE(6,11)C(I,J),(LABEL(I,K),K=1,3)
   11 FORMAT(1H E11.4,3H X 2A4,A2)
   12 SUM=SUM+(C(I,J)**2)/XJ(I)
      F=(DIFF**2)/((XGRPS-1.)*SQUARE*SUM)
   13 WRITE(6,14)F
   14 FORMAT(1H 21X,2HF=E11.4)
   15 STOP
      END
```

Figure 3.—Program listing.

Now, suppose that we want to make contrasts of:

(1) individual treatment response means.

(2) average of responses to treatments 2 and 3 versus average of 4 and 5.

(3) average of responses to treatments 1, 5, 7, and 8 versus average of 2 and 3.

(4) average of responses to treatments 1, 5, 7, and 8 versus average of 4 and 6.

Then the control deck for this job would be that shown in figure 1. The results are shown in figure 2. If we check the calculated F-values[2] against tabulated $F_{(.05)7,134} = 2.08$, we find the following contrasts to be significant:

(1) Treatments 2 and 3, taken separately, versus each of the remaining treatment means (1,4-8).
(2) Contrasts (2), (3), and (4), as stated above.

Figure 3 contains a program listing.

—WARREN E. FRAYER[3]

Assistant Professor of Forest Mensuration
Colorado State University
Fort Collins, Colorado

■

[2] Note that the floating-point multipliers and F-values contain a decimal value, and, after the E, the number of places to the left (—) or right that the decimal point must be moved.

[3] When this paper was prepared, Dr. Frayer was a research forester with the Northeastern Forest Experiment Station, Forest Service, U. S. Department of Agriculture, Upper Darby, Pa.